
Programming
Microsoft Word

Objectives

• Become familiar with some of the key objects in the Microsoft Word
object model.

• Use the Range object to manipulate text in a document.

The Word Object Model

The Word object model is just as sophisticated and extensive as the Microsoft
Excel object model, maybe even more so. Using the Word object model allows
you to control just about any aspect of Word from VBA.

You can begin programming with Word if you understand just a few of the
many objects in the object model. From there, you’ll quickly be able to figure
out what objects, properties, and methods you need to use for a given task.

Since the Word object model is quite complicated, we’ll discuss the
relationships between the different objects in the Word object model as we go
along. To view the entire object model, see Online Help or go to the Microsoft
web site, at http://www.microsoft.com/OfficeDev/Articles/OMG/default.htm.

Word’s Top-Level Objects

The Application Object

The Word Application object represents Word itself and contains all the other
objects and collections in the Word object model. In particular, the Application
object contains two objects you’ll use frequently, the Document and Selection
objects (see Figure 1).

Figure 1. The Application object contains the Documents collection and the
Selection object.

When you’re referring to objects and collections beneath the Application
object, you don’t need to explicitly refer to the Application object. For
example, you can refer to the active document without the Application object.

ActiveDocument.Close

However, when you refer to properties or methods of the Application object
itself, you do need to refer to it explicitly:

Application.CapsLock = True

The Document Object

Just about everything you do in Word will involve the Document object which,
naturally, represents an open document. Document objects belong to the
Documents collection of the Application object (see Figure 1). There are a few
ways to refer to a Document object:

• Refer to the document within the Documents collection, using either
the document’s name or its index.

Application.Documents("vba.doc").Save

• Use the ActiveDocument property. The active document is the one
currently on top in the Word user interface.

ActiveDocument.CheckSpelling

Some Properties and Methods

You can use the Word object model to perform all the tasks you’re accustomed
to when you work with a Word document in the user interface. Here are just a
few examples to get you started.

For example, you can open a document, using the Open method:

Dim doc As Document

Set doc = Documents.Open("vba.doc")

You can create a new document, using the Add method:

Documents.Add

You can print a document in the background, using the PrintOut method:

ActiveDocument.PrintOut Background:=True

You can determine which template is attached to the document, or attach a
new template, using the AttachedTemplate property:

If ActiveDocument.AttachedTemplate = "NORMAL.DOT" Then

ActiveDocument.AttachedTemplate = "MyTemplate.DOT"

End If

You can check a document’s spelling, using the CheckSpelling method:

ActiveDocument.CheckSpelling

You can insert a hyperlink, using the Hyperlinks property:

ActiveDocument.Hyperlinks.Add Selection.Range, _

"http://www.appdev.com"

TIP: If you want to explore Word’s objects, properties, and methods, head to the

Object Browser, and use F1 for context-sensitive Help.

The Selection Object

The Selection object represents the selection in a window. The selection may
be an area of the document, or, if nothing is selected, it may represent the
insertion point.

If you record a macro in Word that involves inserting text, the macro recorder
will often use the Selection object. However, when you’re writing your own
code, you’ll probably want to use the Range object to insert text at a particular
location, and use the Selection object only to work with selected text. We’ll
discuss the Range object in the next section.

You can do a lot with the Selection object. For example, you can copy the
selected text, move the insertion point to a new line, and paste in the text from
the Clipboard:

With Selection

.Copy

.MoveDown

.Paste

End With

Manipulating Text with the Range
Object

What’s a Range?

The Word Range object gives you a way to work with text in a document. It’s
a designated area in your document; a paragraph, several paragraphs, a few
words, a few characters, or even the entire document. Like a Microsoft Excel
range, the Word range can be a variety of shapes and sizes.

A range has a starting position and an ending character position. At the most
granular level, you can define a range by its “coordinates”, which are a row
and column value identifying a particular character’s position in the document.

Referring to a Range

How do you refer to a range? Just like in Microsoft Excel, there’s a Range
property that returns a reference to a Range object. You can then use the
properties and methods of the Range object.

For example, you can define a range in the active document that begins at
position 0 and ends at position 50.

Debug.Print ActiveDocument.Range(0, 50).Text

This code prints the text found in character positions 0 to 50 in the active
document. Notice the Text property, which returns the text in the range.

That’s great, but you probably don’t want to sit in front of your Word
documents counting character positions. Fortunately, you don’t have to. Most
of the objects that represent a part of a document, like a table, a paragraph, or
even a selection, have an associated range you can use.

For example, suppose you want to work with the text of the third paragraph in
the active document. You can use the Range property to get to this text:

Dim strText As String

strText = ActiveDocument.Paragraphs(3).Range.Text

As you can see, in this case you don’t need to know how many characters you
have; instead, you can just refer to a paragraph.

Inserting Text

You can also use the Range object to insert text into a document. It’s simple to
add text to a new document, as the following code shows:

Sub InsertText()

Dim doc As Document

Set doc = Documents.Add

doc.Activate

doc.Range.Text = "Add some text."

End Sub

In this case the Range object refers to the whole document. If this were an
existing document that had text in it already, all of the text would have been
deleted.

What if you want to insert text either before or after other text, without
deleting the original text? You can use the InsertBefore and InsertAfter
methods of a Range object. Let’s take a look at an example.

Try running frmInsertText, in the sample file VBA.doc. Type some text into
the text box, and click Insert Text to insert it into the document as shown in
Figure 2. Play with the settings of the Start of Document and End of Document
options to see how Word inserts the text you typed.

See frmInsertText

Figure 2. Use frmInsertText to insert text into the beginning or end of a Word
document.

How does this form work? Here’s the code in the Click event for the Insert
Text command button:

Private Sub cmdInsert_Click()

Dim doc As Document

Dim intPosition As Integer

' Assign document to object variable.

Set doc = Documents("vba.doc")

' Activate document.

doc.Activate

' Check value of options.

If optStart = True Then

' Use InsertBefore with 0.

doc.Range(0).InsertBefore _

txtInsert.Text

Else

' Store last position in the document.

intPosition = doc.Range.End - 1

' Use InsertAfter with that position.

doc.Range(intPosition).InsertAfter txtInsert.Text

End If

End Sub

When this code runs, it first checks the values of the option buttons to
determine which option the user has chosen. If the user picked Start of
Document, the position to start inserting text in the document is 0. The
procedure calls the InsertBefore method to insert the text before the range
beginning with 0, in other words, at position 0.

If the user selected End of Document, then the code determines the current
ending position for the document, using the End property. You can’t insert text
after the end of the document, so you have to subtract 1 from the value of the
End property. The procedure then calls the InsertAfter method to insert the text
after the last character in the document.

Working with Templates

When you add code to the VBA project associated with a Word document, that
code is available only to that document. However, you might like to write
procedures that you can use with a set of your Word documents, or with all of
them. You can do so by writing code within a template. When you create a
new document based on a template, or when you attach a template to an
existing document, the code in the template is also available to your document.

You have two basic options for packaging your code in a template. You can
write code within the VBA project for the Normal template, the default
template for every Word document, or you can create a custom template.

Using the Normal Template

If you want to write code that is available to every document you create, place
it in the Normal.DOT template. This template is loaded as soon as you start
Word.

The VBA project for the Normal template behaves like any other project. You
may have noticed that whenever you open the Visual Basic Editor, the Normal
project appears in the Project Explorer.

Use the Normal template for the following types of code:

• Generic code to be available to all Word documents. For example, a
procedure that quickly sizes a document and displays it in a particular
view might go in the Normal template.

• Code that manipulates multiple documents at once, or that opens a
document or creates a new document. For example, code that tiles all
open windows vertically on the screen should go in the Normal
template.

Using a Custom Template

If you want to build a custom Word application, it’s best to start with a custom
template. A custom template contains code with related functionality, that you
want to use for a particular set of documents, but that you don’t need to be
available to all documents.

To make the functionality of a custom template available to an individual
document, you can:

• Create the document based on the template. To do this, from the File
menu choose New and select the template you want to use.

• Attach the template to an existing document. From the Tools menu
select Templates and Add-Ins, and click the Attach button to attach
the desired template to the document.

• Include the template as an add-in for the document. From the Tools
menu select Templates and Add-Ins, and click the Add-Ins button to
load the template. Once the template is loaded, any code it contains is
available to the document.

Use a custom template for the following:

• Custom documents that you want to create more than once. For
example, you might create an invoice template, with your company’s
name and address included on the template. The template could
include custom code for performing a mail merge with a customer
database.

• Code that provides a specific functionality to a document. For
example, you could create an add-in making it easy for the user to
mark index entries in a document. You could then load the add-in with
any template or document needing this functionality.

Summary

• The Word object model is complex, but once you understand a few
key objects, you can use it to automate many common tasks in Word.

• The Document object represents an open Word document.

• The Selection object represents the current selection in the document,
or the insertion point, if there is no selection.

• The Range object is useful for working with text by identifying a
position in the document.

• Word Templates can help provide special functionality to new
documents.


